;

;     Функционално програмиране.
;     Упражнение N5

;Задача 1. Да се напише функция за числено интегриране по метода на Симпсън.

(define (sum term a next b)

  (if  (> a b)

       0

       (+ (term a)

          (sum  term (next a) next b))))
(define (integral f a b n)

  (define h (/ (- b a) n))

  (define (next-arg x) (+ x (* 2 h)))

  (* (/ h 3)

     (+ (f a)  (f b)

        (* 4 (sum f (+ a h) next-arg b))

        (* 2 (sum f (+ a (* 2 h)) next-arg (- b h))))))

(define (cube x) (* x x x))
(define (one x) 1)

(define (same x) x)

;(integral one 0. 1. 100) -> 1.
;(integral same 0. 1. 100) -> 0.5 

;(integral one 1. 9. 8) -> 8.
;Итеративна форма на sum

;(define (sum term a next b)

;  (define (iter a result)

;    (if (> a b)

;        result

;        (iter (next a)

;              (+ (term a) result))))

;    (iter a 0))

;Задача 2. Да се напише функция product, аналогична на

;(sum term a next b), която връща произведението на стойности

;на функция в точки от даден интервал. Функцията да се напише

;в две форми - едната да генерира рекурсивен процес, а другата -

;итеративен процес. Как да се дефинира factorial в термините на

;product? Да се използва product за пресмятане на П съгласно

;PI/2 = 2/1 * 2/3 * 4/3 * 4/5 * 6/5 *...
;За използване на П в DrScheme се зарежда библиотека с require
;(require (lib "math.ss"))

;pi -> 3.141592653589793

;product чрез линейна рекурсия

(define (product term a next b)

  (if (> a b)

      1

      (* (term a) (product term (next a) next b))))

;product чрез линейна итерация

(define (product-it term a next b)

  (define (iter a result)

    (if (> a b) result

        (iter (next a) (* result (term a)))))

  (iter a 1))

(define (fact n)

  (define (same x) x)

  (define (inc x) (+ x 1))

  (product same 1 inc n))

(define (product-pi a b)

  (define (pi-term x)

    (/ (square x) (* (- x 1) (+ x 1))))

  (define (next x) (+ x 2))

  (product pi-term a next b))
(define (square x) (* x x))
;(product-pi 2 100) -> 1.563

;(product-pi 2 1000) -> 1.57

;Задача 3. Покажете, че sum  и product са специални случаи на още ;по-обща идея, наречена accumulate, която комбинира съвкупност от ;терми, използвайки някаква обща акумулираща функция:

;(accumulate combiner null-value term a next b)

;accumulate има за аргументи същия терм и спецификации за начало и

;край на интервала като sum и product, функция combiner (на два

;аргумента), която специфицира как текущия терм да бъде комбиниран

;с натрупването на предишните терми и null-value, която специфицира

;каква начална стойност да се използва. Напишете accumulate (в

;рекурсивна и итеративна форма) и покажете как sum и product могат

;да бъдат дефинирани като прости извиквания на accumulate.

;accumulate с линейна рекурсия.

(define (inc x) (+ x 1))

(define (same x) x)

(define (accumulate combiner null-value term a next b)

  (if (> a b)

      null-value

      (combiner (term a)

                (accumulate combiner null-value term (next a) next b))))

(define (sum term a next b)

  (accumulate + 0 term a next b))

(define (product term a next b)

  (accumulate * 1 term a next b))

;accumulate с линейна итерация.

(define (accumulate-it combiner null-value term a next b)

  (define (iter a result)

    (if (> a b)

        result

        (iter (next a)

              (combiner result (term a)))))

  (iter a null-value))

;Задача 4. Да се напише по-обща версия на accumulate чрез въвеждане ;на филтър (filter) на термовете, които да бъдат комбинирани. Това ;означава, че няма да се комбинират всички термове в интервала, а ;само тези, които удовлетворяват посочено условие. Функцията filter-;accumulate има същите аргументи като accumulate с допълнителен ;предикат на един аргумент, който специфицира филтъра. Да се покаже ;как да се изрази чрез filter-accumulate:

;a) сумата от квадратите на простите числа в интервала [a,b]

;(предикатът prime? вече е написан);

;(sum-square 1 10) -> 88

;б) произведението на всички положителни цели числа a < n, такива,

;че GCD(a,n) = 1 (a и n са взаимно прости).

;(prod-gcd 10) -> 189  3.7.9 = 189

;filter-accumulate с линейна рекурсия.

(define (filter-accumulate combiner null-value term a next b filter?)

  (cond ((> a b) null-value)

        ((filter? a)

         (combiner

           (term a)

           (filter-accumulate

             combiner null-value term (next a) next b filter?)))

        (else

          (filter-accumulate

             combiner null-value term (next a) next b filter?))))
(define (same x) x)

(define (inc x) (+ x 1))

(define (sum-square a b)

  (filter-accumulate + 0 square a inc b prime?))

(define (prod-gcd n)

  (define (gcd? a) (= (gcd n a) 1))

  (filter-accumulate * 1 same 2 inc n gcd?))

;filter-accumulate с линейна итерация.

(define (filter-accumulate combiner null-value term a next b filter?)

  (define (iter a result)

    (cond ((> a b) result)

          ((filter? a) (iter (next a) (combiner result (term a))))

          (else (iter (next a) result))))

  (iter a null-value))

